1. Shocks from a Coefficient

It is now possible to take the values of shocks from the values of a Coefficient. This possibility was introduced in Release 10.0-002 (April 2010) of GEMPACK.

The two new statements are of the form

shock <var-name>[(<var-args)] = 
  coefficient <coeff-name>[(coeff-args)] ;

shock <var-name>[(<var-args)] = select from
  coefficient <coeff-name>[(coeff-args)]

Here both <var-args> and <coeff-args> are optional.

Note that the first word can be any of

shock, ashock, tshock, change, achange, tchange, 
Percent_Change, APercent_change, TPercent_Change,.
Final_Level or TFinal_Level.

[See section 7.7 of GPD-9 for many of these.]

All of these are similar to the case where shock values are read from a Header Array file (see section 5.5.2 of GPD-3).

In the "select from" case, the part of the Coefficient specified after the word "coefficient" must range over the same sets as the whole of the variable specified in the shock statement. The parts of the variable on the LHS get shocks from the corresponding parts of the Coefficient on the RHS.

When "select from" is not present, the part of the Coefficient specified after the word "coefficient" must range over the same sets as the part of the variable specified in the shock statement.

We give precise rules later.

1.1.1 Examples

Consider a model in which there are sets COM, IND, COM2 and IND2 with COM2 a subset of COM and IND2 a subset of IND. Suppose that Coefficients and Variables are declared via the statements

Coefficient (all,c,COM) Coef1(c) ;
Coefficient (all,c,COM)(All,i,IND) Coef2(c,i) ;
Variable (all,c,COM) v1(c) ;
Variable (all,i,IND) v2(i) ;

Then the following statements are allowed.

Shock v1 = coefficient Coef1 ;

Shock v1(COM2) = select from coefficient Coef1 ;

Shock v1(COM2) = coefficient Coef1(COM2) ;

Shock v1(COM2) = select from coefficient Coef2(COM, "i3") ;

Shock v1(COM2) = coefficient Coef2(COM2) ;

Shock v2 = coefficient Coef2("c2", IND) ;

Shock v2(IND2) = coefficient Coef2("c5", IND2) ;

Shock v2(IND2) = select from coefficient Coef2("c1", IND) ;

In each case the obvious shocks are given to the relevant components of variable v1 or v2.

1.1.2 Motivation

In many cases, "shock-from-coefficient" statements in your CMF file may eliminate the need for a separate program (coded in Excel or in TABLO language) that calculates shocks. Instead the code to calculate shocks may be incorporated in the main model file, which probably will already include much of the apparatus that is needed. Thus "shock-from-coefficient" can reduce the number of programs that need to be maintained and checked. This is especially true where the shock can be computed from standard initial data (eg, shocks to powers of taxes that halve the ad valorem rate of tax).

"Shock-from-coefficient" statements are also very handy for recursive-dynamic models run using RunDynam – which repeatedly executes your model program, each simulation starting from an updated database produced by the previous simulation. Suppose you wished to shock such a model with values that depended on the data for the current year. You could not calculate shocks for all years before running the experiment. And RunDynam does not allow you to run a separate shock-calculating program before each year's simulation. "Shock-from-coefficient" statements avoid this problem: the model itself calculates and applies shocks for each year.

Similarly, stock-flow accumulation and partial-adjustment relations that often occur in dynamic models may often be coded more simply by using "Shock-from-coefficient" statements
.

As well, "shock-from-coefficient" statements are more reliable than reading shocks from an external file. For example, if you read shocks from a text file it is quite possible that the shocks on file are ordered incorrectly (eg, regions or commodities in wrong order). GEMSIM or a TABLO-generated program probably would not detect such an error – you need to carefully check. That type of error is eliminated if you use "shock-from-coefficient" statements.

1.1.3 Formal Rules

When deciding whether or not one of these statements is valid,

we ignore elements and sets of size 1.

The examples below (which use the sets, Coefficients and Variables in section 1.1.1 above) will make this clear.

In these rules, 

· if <var-args> are not present, they are taken to be the full sets over which the Variable is declared (in the TAB file).

· if <coeff-args> are not present, they are taken to be the full sets over which the Coefficient is declared (in the TAB file). 

For example, 

shock v1 = coefficient coef1 ;

is the same as

shock v1(COM) = coefficient coef1(COM) ;

1. Consider firstly statements which do not have "select from".

shock <var-name>[(<var-args)] = 
  coefficient <coeff-name>[(coeff-args)] ;

Then the sets in <var-args> must be the same sets (indeed, have the same names)
 as those in <coeff-args>, ignoring elements and sets of size 1.

For example, the following are not valid.

shock v1(COM2) = coefficient coef1 ;   
   ! invalid. <coeff-args> are COM, different from COM2

shock v1 = coefficient coef2(COM, IND) ;
   ! invalid. 2 sets in <coeff-args> only 1 in <var-args>

If COM3 is another subset of COM of the same size as COM2, then

shock v1(COM3) = coefficient coef1(COM2) ; ! invalid

is invalid since, although there are the same number of components on the LHS as there are coefficient values on the RHS, the set COM2 is not the same set as COM3 (even if it has the same elements as COM3).

2. Consider firstly statements which have "select from".

shock <var-name>[(<var-args)] = select from
  coefficient <coeff-name>[(coeff-args)] ;

Then the sets over which var-name is declared in the TAB file must be the same sets (indeed, have the same names)
 as those in <coeff-args>, ignoring elements and sets of size 1. 

For example (with sets etc as in section 1.1.1),

shock v1(COM2) = select from coefficient coef1 ; ! For each element <elt> in
                              ! COM2, the shock to v1(<elt>) is equal to coef1(<elt>).
Notice here that any <var-args> specified are irrelevant as to whether or not this statement is valid.

For example, the following are not valid.

shock v1(COM2) = select from coefficient coef1(COM2) ;
  ! invalid.

shock v1(COM2) = select from coefficient 
    coef2(COM2, "i3") ; ! invalid

If there are several sets involved in the arguments, you can still see what is happening by ignoring elements and sets of size 1.

For example, suppose that we also have

Coefficient (all,c,com)(all,s,SRC)(all,i,IND)(all,m,MAR)
     Coef3(c,s,i,m) ;

Variable (all,c,com)(all,s,SRC)(all,i,IND) v3(c,s,i) ;

where SRC = (dim, imp) and MAR is a set with just a single element. Then the statements

shock v2(IND2) = select from 
      coefficient Coef3("c2", "imp", IND, MAR) ;

shock v3(COM,IND2) = coefficient Coef3(COM, "dom", IND2, MAR) ;

are both valid.

1.1.4 Fine Print

In a shock statement with "coefficient" on the RHS, you are not allowed to specify component numbers of the variable. For example

Shock v1 2-4 = coefficient Coef1(COM) ; 
     ! invalid. Component numbers not allowed

is invalid.

The values of the Coefficient used are the pre-simulation values of that Coefficient – that is, the values worked out during the first pass of the multi-step calculation.

The coefficient appearing after the word "coefficient" must not be a post-sim Coefficient (since its values will not be available when the shock statement is processed).

The coefficient appearing after the word "coefficient" must not be a system-initiated Coefficient (one introduced by TABLO during Condensation – see section 2.3.2 of GPD-2).

The sets in <var-args> and <coeff-args> do not have to have their elements in increasing order. [Contrast this with section 5.5.3 of GPD.3.]

� Both stock-flow and partial-adjustment relations can cause changes in model variables even if no external shock is applied. That is, they include a disequilibrium term, akin to a shock. In older TABLO code, such equations often include an ordinary change variable called (probably) delUnity – which is always exogenous and always shocked by 1. The delUnity variable is multiplied by a coefficient which may be interpreted as an ordinary change shock. Additional coding tricks might be needed to convert that ordinary change shock into a percentage change. Often, "shock-from-coefficient" statements could be used to express such relations more clearly.


� Just having the same elements is not sufficient – the sets must be identical (in name). For example, the statement�Shock v1(COM2) = coefficient Coef1(COM3) ;�is not valid even if COM2 and COM3 are declared to be equal sets in the TAB file.


� Just having the same elements is not sufficient – the sets must be identical (in name). For example, the statement�shock v1(COM2) = select from coefficient coef3 ;�is invalid if COEF3 is defined over a set COM6, even if COM6 and COM are declared to be equal sets in the TAB file.





